

CSE301(Functional Programming)  
Lecture 2: Higher-order functions and type classes

Noam Zeilberger

version: September 13, 2023

A **higher-order function** is a function that takes one or more functions as input. A primary motivation for defining a higher-order function is to express the common denominator between a collection of first-order functions. Higher-order functional programming is thus a way to promote code reuse. Although it may take you a bit more time to understand what a higher-order function is doing than the corresponding first-order ones, eventually it will help you in “seeing the forest for the trees”, and hopefully give you insights into solving certain kinds of programming problems much more quickly.

## 1 0th example: applying a function twice

We can write a simple higher-order function that takes a function and applies it twice to some argument.

$$\begin{aligned} \textit{twice} &:: (a \rightarrow a) \rightarrow a \rightarrow a \\ \textit{twice} \ f \ x &= f \ (f \ x) \end{aligned}$$

For example, we have<sup>1</sup>

$$\begin{aligned} \textit{twice} \ (1+) \ 3 &= 5 \\ \textit{twice} \ (\backslash n \rightarrow n * n) \ 3 &= 81 \\ \textit{twice} \ \textit{reverse} \ [1, 2, 3] &= [1, 2, 3] \end{aligned}$$

In the rest of the lecture we will be considering many more interesting examples.

## 2 First example: abstracting case-analysis

As we discussed in Lecture 1, sum types obey the general principle that if  $f :: a \rightarrow c$  and  $g :: b \rightarrow c$  are two functions taking types  $a$  and  $b$  to the same target type  $c$ , then there is a canonical function

$$\begin{aligned} h &:: \textit{Either} \ a \ b \rightarrow c \\ h \ (\textit{Left} \ x) &= f \ x \\ h \ (\textit{Right} \ y) &= g \ y \end{aligned}$$

---

<sup>1</sup>In Haskell, the expression  $(1+)$  is called a (left) *section* of the infix operator  $(+)$ . It is equivalent to the lambda expression  $\backslash n \rightarrow 1 + n$ . The lambda expression  $\backslash n \rightarrow n * n$  denotes the function which squares its argument.

taking the sum type to the same target, defined by pattern-matching on the *Left* or *Right* constructor – or in other words, by case-analysis. Here again are two concrete examples of functions defined by case-analysis on *Either Bool Int*:

```

asInt :: Either Bool Int → Int
asInt (Left b) = if b then 1 else 0
asInt (Right n) = n
isBool :: Either Bool Int → Bool
isBool (Left b) = True
isBool (Right n) = False

```

The Haskell Prelude defines a higher-order function that “internalizes” the principle of case-analysis over sum types, so to speak, by taking the functions  $f$  and  $g$  as extra arguments:

```

either :: (a → c) → (b → c) → Either a b → c
either f g (Left x) = f x
either f g (Right y) = g y

```

Here is how we can redefine *asInt* and *isBool* using *either* and lambda notation:

```

asInt = either (\b → if b then 1 else 0) (\n → n)
isBool = either (\b → True) (\n → False)

```

Whereas before we could spot that the two functions were instances of a simple common “design pattern”, now they are literally two applications of the same (higher-order) function.

Observe that here we have only partially applied the function *either* to a pair of functions  $f$  and  $g$ , to compute new functions *either f g*. Alternatively, we could have defined

```

asInt v = either (\b → if b then 1 else 0) (\n → n) v
isBool v = either (\b → True) (\n → False) v

```

supplying an extra argument  $v :: Either Bool Int$ , but in Haskell these two versions are completely equivalent. (They are said to be  $\eta$ -equivalent, in lambda calculus jargon.) Finally, recall that the arrow type constructor associates to the right by default, so that the type of *either* may be equivalently written with extra parentheses as:

```

either :: (a → c) → ((b → c) → (Either a b → c))

```

In this form it looks a lot like the logical formula

$$(A \supset C) \supset ([B \supset C] \supset [(A \vee B) \supset C])$$

which you can easily verify is a tautology. And indeed this analogy can be made precise, with many examples of higher-order functions corresponding to proofs of nested implications.

### 3 Second example: mapping over a list

Consider the following first-order functions on lists:

**(Add one to every element in a list of integers.)**

```
mapAddOne :: [Integer] → [Integer]
mapAddOne [] = []
mapAddOne (x : xs) = (1 + x) : mapAddOne xs
```

Example: `mapAddOne [1..5] = [2,3,4,5,6]`

**(Square every element in a list of integers.)**

```
mapSquare :: [Integer] → [Integer]
mapSquare [] = []
mapSquare (x : xs) = (x * x) : mapSquare xs
```

Example: `mapSquare [1..5] = [1,4,9,16,25]`

**(Compute the length of each list in a list of lists.)**

```
mapLength :: [[a]] → [Int]
mapLength [] = []
mapLength (x : xs) = length x : mapLength xs
```

Example: `mapLength ["hello", "world!"] = [5,6]`

Clearly there is a common denominator here of “apply some transformation to every element of a list”. And again, this design pattern can be internalized as a higher-order function, which is called *map* in the Prelude:

```
map :: (a → b) → [a] → [b]
map f [] = []
map f (x : xs) = (f x) : map f xs
```

For example, we have:

```
mapAddOne = map (1+)
mapSquare = map (\n → n * n)
mapLength = map length
```

## 4 Some useful functions on functions

As we’ve talked about before, functions of multiple arguments are usually defined in “curried” form in Haskell, that is, by giving them a right-nested function type – although it is equally possible to define them as functions taking a product type as input, and these two forms are equivalent. Indeed, the “currying principle” may be internalized as a higher-order function:

```
curry :: ((a, b) → c) → (a → b → c)
curry f x y = f (x, y)
```

which turns a function on pairs into a curried function, and conversely, there is an “uncurrying” principle:

$$\begin{aligned} \text{uncurry} &:: (a \rightarrow b \rightarrow c) \rightarrow ((a, b) \rightarrow c) \\ \text{uncurry } g &(x, y) = g \ x \ y \end{aligned}$$

which turns a curried function into a function on pairs. Note that both of these may be equivalently written using lambda notation:

$$\begin{aligned} \text{curry } f &= \backslash x \rightarrow \backslash y \rightarrow f \ (x, y) \\ \text{uncurry } g &= \backslash (x, y) \rightarrow g \ x \ y \end{aligned}$$

One reason we may want to apply currying or uncurrying in practice is to transform a function before providing it as an argument to another higher-order function. For example,  $\text{map } (\text{uncurry } (+)) [(0, 1), (2, 3), (4, 5)] = [1, 5, 9]$ . Similarly to what we saw with the higher-order function *either*, the types of the currying and uncurrying transformations can be read as expressing a logical equivalence, namely that

$$(A \wedge B) \supset C \iff A \supset (B \supset C)$$

which again you can easily verify is a tautology.

Another simple but fundamental principle for defining functions is the *principle of sequential composition*: given a function  $f :: a \rightarrow b$  and a function  $g :: b \rightarrow c$  there is a function  $h :: a \rightarrow c$  defined by  $h \ x = g \ (f \ x)$ . Once more, this principle may be expressed as a higher-order function:

$$\begin{aligned} (\circ) &:: (b \rightarrow c) \rightarrow (a \rightarrow b) \rightarrow (a \rightarrow c) \\ (g \circ f) \ x &= g \ (f \ x) \end{aligned}$$

(Or equivalently:  $g \circ f = \backslash x \rightarrow g \ (f \ x)$ .) For example,  $\text{map } ((+1) \circ (*2)) [0..4] = [1, 3, 5, 7, 9]$ . Note that in Haskell, composition is actually notated using a period ( $g \ . \ f$ ), although we use the notation  $g \circ f$  here in these notes to emphasize the relation to the standard mathematical concept. Logically, the type of function composition corresponds to the principle of *transitivity of implication*.

Let us conclude by describing a few more important higher-order functions:<sup>2</sup>

**(Flip the order in which a binary function takes its arguments.)**

$$\begin{aligned} \text{flip} &:: (a \rightarrow b \rightarrow c) \rightarrow (b \rightarrow a \rightarrow c) \\ \text{flip } f \ x \ y &= f \ y \ x \end{aligned}$$

Example:  $\text{flip } (-) \ 2 \ 3 = 1$

**(Given a value, build the constant function returning that value.)<sup>3</sup>**

$$\begin{aligned} \text{const} &:: b \rightarrow (a \rightarrow b) \\ \text{const } x \ y &= x \end{aligned}$$

Example:  $\text{const } 2 \ 3 = 2$

**(Turn a binary function into a unary function duplicating its input.)**

$$\begin{aligned} \text{dupl} &:: (a \rightarrow a \rightarrow b) \rightarrow (a \rightarrow b) \\ \text{dupl } f \ x &= f \ x \ x \end{aligned}$$

Example:  $\text{dupl } (*) \ 3 = 9$

<sup>2</sup>Both *flip* and *const* are already defined in the Haskell Prelude, but not *dupl*. Observe that the types of all three may be read as basic logical tautologies involving implication.

<sup>3</sup>Note that *const* can also be used to turn unary function into a binary function, and more generally a function of  $n$  arguments into a function of  $n + 1$  arguments that ignores its first argument.

## 5 More higher-order functions on lists

The Haskell Prelude and Standard Library define a number of higher-order functions that capture common ways of manipulating lists. We describe a few of them here.

**(Keep only those elements of a list satisfying a predicate.)**

```
filter :: (a → Bool) → [a] → [a]
filter p [] = []
filter p (x : xs)
  | p x = x : filter p xs
  | otherwise = filter p xs
```

Examples:

```
> filter (>3) [1..5]
[4,5]
> filter Data.Char.isUpper "Glasgow Haskell Compiler"
"GHC"
```

**(Test whether all/any elements of a list satisfy a predicate, and find an element that does.)**

```
all, any :: (a → Bool) → [a] → Bool
all p [] = True
all p (x : xs) = p x && all p xs
any p [] = False
any p (x : xs) = p x || any p xs
find :: (a → Bool) → [a] → Maybe a
find p [] = Nothing
find p (x : xs)
  | p x = Just x
  | otherwise = find p xs
```

Examples:

```
> all (>3) [1..5]
False
> any (>3) [1..5]
True
> find (>3) [1..5]
Just 4
```

**(Return the longest prefix of a list of elements that satisfy a predicate, or the remaining suffix after this prefix is dropped.)**

```
takeWhile, dropWhile :: (a → Bool) → [a] → [a]
takeWhile p [] = []
takeWhile p (x : xs)
  | p x = x : takeWhile p xs
```

```

    | otherwise = []
dropWhile p [] = []
dropWhile p (x : xs)
    | p x       = dropWhile p xs
    | otherwise = x : xs

```

Examples:

```

> takeWhile (>3) [1..5]
[]
> takeWhile (<3) [1..5]
[1,2]
> dropWhile (>3) [1..5]
[1,2,3,4,5]
> dropWhile (<3) [1..5]
[3,4,5]

```

**(Map a function sending an element to a list over all the elements of a list, and concatenate the results.)**

```

concatMap :: (a -> [b]) -> [a] -> [b]
concatMap f [] = []
concatMap f (x : xs) = f x ++ concatMap f xs

```

Examples:

```

> concatMap (\x -> [x]) [1..5]
[1,2,3,4,5]
> concatMap (\x -> if x `mod` 2 == 1 then [x] else []) [1..5]
[1,3,5]
> concatMap (\x -> concatMap (\y -> [x..y]) [1..3]) [1..3]
[1,1,2,1,2,3,2,2,3,3]

```

Note that  $\text{concatMap } f = \text{concat} \circ \text{map } f$ , where  $\text{concat} :: [[a]] \rightarrow [a]$  computes the concatenation of a list of lists.

The higher-order function  $\text{concatMap}$  may be used to understand “list comprehension” notation. For example, the expression

$$[(x, y) \mid x \leftarrow [1..3], y \leftarrow [1..3], x < y]$$

which evaluates to  $[(1, 2), (1, 3), (2, 3)]$ , may be equivalently expressed using  $\text{concatMap}$  as

```

concatMap (\x -> concatMap (\y ->
    if x < y then [(x, y)] else []) [1..3]) [1..3]

```

or equivalently (making use of the higher-order function  $\text{flip}$ ) as

```

flip concatMap [1..3] (\x ->
flip concatMap [1..3] (\y ->
if x < y then [(x, y)] else []))

```

( $\text{concatMap}$  is also related to the monad structure on lists, which this translation of list comprehensions implicitly relies upon. We will talk more about that in Lecture 4.)


## 6 *foldr*: the Swiss army knife of list functions

Remarkably, all of the higher-order functions on lists that we described in Section 5, and many other functions besides, can be defined as instances of a single higher-order function. Before jumping to the definition of this higher-order function, let's give a bit of motivation of where the definition comes from.

Suppose that we want to write a function that transforms a list of *as* into a value of type *b*. Moreover, suppose that we want the function to be defined *inductively*, in the sense that we will provide a base case for the value of the function on the empty list [], and we will explain how to inductively compute the value of the function on a non-empty list  $x :: xs$  from *x* and the value of the function on *xs*. To make this situation a bit more precise, let's give a name to the function,  $h :: [a] \rightarrow b$ . To “provide a base case” means that we provide a value  $v :: b$  such that  $h [] = v$ , while to “explain how to inductively compute the value of the function on a non-empty list” means that we provide a function  $f :: a \rightarrow b \rightarrow b$  such that  $h (x :: xs) = f x (h xs)$ . We thus obtain the following recursive definition of *h*:

$$\begin{aligned} h &:: [a] \rightarrow b \\ h [] &= v \\ h (x : xs) &= f x (h xs) \end{aligned}$$

The schema for defining *h* can be summarized by the following diagram (similar to the coproduct diagram we saw in the previous lecture):



Since this schema for defining *h* is completely generic in the “base case” *v* and the “inductive step” *f*, we can internalize it as a higher-order function, which is called *foldr* in the Haskell Prelude:

$$\begin{aligned} foldr &:: (a \rightarrow b \rightarrow b) \rightarrow b \rightarrow [a] \rightarrow b \\ foldr f v [] &= v \\ foldr f v (x : xs) &= f x (foldr f v xs) \end{aligned}$$

Here are definitions for most of the functions in Section 5 in terms of *foldr*:

$$\begin{aligned} filter\ p &= foldr (\backslash x\ xs \rightarrow \mathbf{if}\ p\ x\ \mathbf{then}\ x : xs\ \mathbf{else}\ xs)\ [] \\ all\ p &= foldr (\backslash x\ b \rightarrow p\ x\ \&\&\ b)\ True \\ any\ p &= foldr (\backslash x\ b \rightarrow p\ x\ ||\ b)\ False \\ find\ p &= foldr (\backslash x\ mx \rightarrow \mathbf{if}\ p\ x\ \mathbf{then}\ Just\ x\ \mathbf{else}\ mx)\ Nothing \\ takeWhile\ p &= foldr (\backslash x\ xs \rightarrow \mathbf{if}\ p\ x\ \mathbf{then}\ x : xs\ \mathbf{else}\ [])\ [] \\ concatMap\ f &= foldr (\backslash x\ ys \rightarrow f\ x\ ++\ ys)\ [] \end{aligned}$$

And here are a few more examples of first-order functions defined by pattern-matching and recursion, and their more concise definitions in terms of *foldr*:

**(Take the sum of the numbers in a list.)**

$$\begin{aligned} \text{sum} &:: \text{Num } a \Rightarrow [a] \rightarrow a \\ \text{sum } [] &= 0 \\ \text{sum } (x : xs) &= x + \text{sum } xs \end{aligned}$$

which may be summarized as:

$$\text{sum} = \text{foldr } (+) 0$$

**(Take the product of the numbers in a list.)**

$$\begin{aligned} \text{product} &:: \text{Num } a \Rightarrow [a] \rightarrow a \\ \text{product } [] &= 1 \\ \text{product } (x : xs) &= x * \text{product } xs \end{aligned}$$

which may be summarized as:

$$\text{product} = \text{foldr } (*) 1$$

**(Return the length of a list.)**

$$\begin{aligned} \text{length} &:: [a] \rightarrow \text{Int} \\ \text{length } [] &= 0 \\ \text{length } (x : xs) &= 1 + \text{length } xs \end{aligned}$$

which may be summarized as:

$$\text{length} = \text{foldr } (\backslash x n \rightarrow 1 + n) 0 = \text{foldr } (\text{const } (1+)) 0$$

**(Return the concatenation of a list of lists.)**

$$\begin{aligned} \text{concat} &:: [[a]] \rightarrow [a] \\ \text{concat } [] &= [] \\ \text{concat } (xs : xss) &= xs \text{ ++ } \text{concat } xss \end{aligned}$$

which may be summarized as:

$$\text{concat} = \text{foldr } (++) []$$

**(Copy a list.)**

$$\begin{aligned} \text{copy} &:: [a] \rightarrow [a] \\ \text{copy } [] &= [] \\ \text{copy } (x : xs) &= x : \text{copy } xs \end{aligned}$$

which may be summarized as:

$$\text{copy} = \text{foldr } (:) []$$



In all of the above examples we used *foldr* to define a function taking a single list as input. However, it is possible to use *foldr* to define functions of multiple arguments, by seeing them as curried functions, and taking the target type *b* in the type schema

$$\text{foldr} :: (a \rightarrow b \rightarrow b) \rightarrow b \rightarrow [a] \rightarrow b$$

to be a function type. For example, consider the binary concatenation operation  $(++)$ , which we saw how to define recursively in Lecture 1:

$$\begin{aligned} (++) &:: [a] \rightarrow [a] \rightarrow [a] \\ [] ++ ys &= ys \\ (x : xs) ++ ys &= x : (xs ++ ys) \end{aligned}$$

To translate this into a fold, let's first get rid of the infix notation and write  $(++)$  as a prefix operator, just so as to not get distracted by syntax:

$$\begin{aligned} (++) &:: [a] \rightarrow [a] \rightarrow [a] \\ (++) [] ys &= ys \\ (++) (x : xs) ys &= x : ((++) xs ys) \end{aligned}$$

And then let's write the function equivalently using lambda notation, while also inserting a judicious pair of parentheses into its type:

$$\begin{aligned} (++) &:: [a] \rightarrow ([a] \rightarrow [a]) \\ (++) [] &= \backslash ys \rightarrow ys \\ (++) (x : xs) &= \backslash ys \rightarrow x : ((++) xs ys) \end{aligned}$$

We can now see how to fit  $(++)$  within the general inductive schema for defining functions  $h :: [a] \rightarrow b$  described at the beginning of this section:

1. We take  $b = [a] \rightarrow [a]$  as the target type.
2. We take the identity function  $\backslash ys \rightarrow ys$  for the base case  $v :: [a] \rightarrow [a]$ .
3. We take the higher-order function  $\backslash x g ys \rightarrow x : g ys$  for the inductive case  $f :: a \rightarrow ([a] \rightarrow [a]) \rightarrow ([a] \rightarrow [a])$ .

Finally, writing *id* for the identity function, and observing that

$$(\backslash x g ys \rightarrow x : g ys) = (\backslash x g \rightarrow (x:) \circ g)$$

we derive the following concise definition of  $(++)$ :

$$(++) = \text{foldr } (\backslash x g \rightarrow (x:) \circ g) \text{ id}$$

**Exercise 6.1.** Use *foldr* to define a function  $\text{altsum} :: \text{Num } a \Rightarrow [a] \rightarrow a$  which takes the alternating sum of the numbers in a list. (Example:  $\text{altsum } [1..5] = 1 - 2 + 3 - 4 + 5 = 3$ .)

**Exercise 6.2.** Use *foldr* to define the following functions from the standard library exported by `Data.List`:

1.  $\text{intersperse} :: a \rightarrow [a] \rightarrow [a]$  which intersperses a given element between the elements of a list. (Example:  $\text{intersperse } ', ' "abcde" = "a,b,c,d,e"$ .)

2.  $\text{tails} :: [a] \rightarrow [[a]]$  that returns the list of all suffixes of a list. (Example:  $\text{tails} \text{ "abcde"} = [\text{ "abcde"}, \text{ "bcde"}, \text{ "cde"}, \text{ "de"}, \text{ "e"}, \text{ ""}]$ .)
3.  $\text{isPrefixOf} :: Eq a \Rightarrow [a] \rightarrow [a] \rightarrow Bool$  determining whether the first list is a prefix of the second. (Examples:  $\text{isPrefixOf} \text{ "abc"} \text{ "abcde"} = True$  but  $\text{isPrefixOf} \text{ "abc"} \text{ "def"} = False$ .)

**Exercise 6.3.** The astute reader will have noticed that we did not define *dropWhile* when we showed how to define the higher-order functions from Section 5 in terms of *foldr*. Nevertheless it is possible to define *dropWhile p* just using *foldr*, without using recursion. Explain how. *Hint:* Be careful not to define a variation of *filter*. You are allowed to invoke *foldr* and then do something with the result, as long as you do not use recursion. Note that there is more than one solution to this problem.

## 7 Aside: Folding from the left

Let's have a closer look at the way that the definition of *sum* in terms of *foldr* is used to compute the sum of the numbers 1 through 5:

$$\begin{aligned}
 \text{sum} [1..5] &= \text{foldr} (+) 0 [1, 2, 3, 4, 5] \\
 &= 1 + \text{foldr} (+) 0 [2, 3, 4, 5] \\
 &= 1 + (2 + \text{foldr} (+) 0 [3, 4, 5]) \\
 &= 1 + (2 + (3 + \text{foldr} (+) 0 [4, 5])) \\
 &= 1 + (2 + (3 + (4 + \text{foldr} (+) 0 [5]))) \\
 &= 1 + (2 + (3 + (4 + (5 + \text{foldr} (+) 0 [])))) \\
 &= 1 + (2 + (3 + (4 + (5 + 0)))) \\
 &= 1 + (2 + (3 + (4 + 5))) \\
 &= 1 + (2 + (3 + 9)) \\
 &= 1 + (2 + 12) \\
 &= 1 + 14 \\
 &= 15
 \end{aligned}$$

We see that the additions are performed from right to left. In general, *foldr f v xs* computes its answer by starting with the value *v* and combining it using *f* with each of the elements of *xs*, going right-to-left from the last element to the first.

Sometimes it can be more natural and/or more efficient to go in the other direction, applying the operation to the elements of the list from left to right. This general pattern of recursion is captured by the higher-order function *foldl*:

$$\begin{aligned}
 \text{foldl} &:: (b \rightarrow a \rightarrow b) \rightarrow b \rightarrow [a] \rightarrow b \\
 \text{foldl } f \ v \ [] &= v \\
 \text{foldl } f \ v \ (x : xs) &= \text{foldl } f \ (f \ v \ x) \ xs
 \end{aligned}$$

Many of the functions we defined using *foldr* in Section 6 may also be defined using *foldl*, for instance:

$$\begin{aligned}
 \text{sum} &= \text{foldl} (+) 0 \\
 \text{product} &= \text{foldl} (*) 1 \\
 \text{concat} &= \text{foldl} (++) []
 \end{aligned}$$

These definitions are equivalent to the ones using *foldr* in the sense that they yield the same functions from inputs to output, but in fact the recursive computations are very different. For example, here is the computation of the sum of the numbers 1 through 5 using *foldl*:

```

foldl (+) 0 [1, 2, 3, 4, 5]
= foldl (+) 1 [2, 3, 4, 5]
= foldl (+) 3 [3, 4, 5]
= foldl (+) 6 [4, 5]
= foldl (+) 10 [5]
= foldl (+) 15 []
= 15

```

We see that now the additions are evaluated from left to right. Of course the order of evaluation does not matter for the final result here because addition is an associative operation, although it may matter for efficiency.

**Exercise 7.1.** The computation of *foldl* (+) 0 [1, 2, 3, 4, 5] might remind you of the implementation of *reverse* using an accumulator that we discussed in Lecture 1. Show how to define *reverse* just using *foldl*.

**Exercise 7.2.** Prove that if  $f :: a \rightarrow a \rightarrow a$  and  $v :: a$  satisfy the equations of a monoid, then  $foldr f v = foldl f v$ .

## 8 Higher-order functions over trees

We've spent a while discussing higher-order functions over lists, but it is often useful to define higher-order functions over other data types as well. For instance, recall the data type of binary trees with labelled nodes that we introduced in Lecture 1:

```

data BinTree a = Leaf | Node a (BinTree a) (BinTree a)
deriving (Show, Eq)

```


It supports a natural analogue of the *map* function on lists, applying some transformation to every node label contained in the tree:

```

mapBT :: (a → b) → BinTree a → BinTree b
mapBT f Leaf = Leaf
mapBT f (Node x tL tR) = Node (f x) (mapBT f tL) (mapBT f tR)

```

For example, we have:



There is also a natural analogue of *foldr*, which inductively defines a function  $BinTree\ a \rightarrow b$  by providing a “base case” and an “inductive step”:

$$\begin{aligned} foldBT &:: b \rightarrow (a \rightarrow b \rightarrow b \rightarrow b) \rightarrow BinTree\ a \rightarrow b \\ foldBT\ v\ f\ Leaf &= v \\ foldBT\ v\ f\ (Node\ x\ tL\ tR) &= f\ x\ (foldBT\ v\ f\ tL)\ (foldBT\ v\ f\ tR) \end{aligned}$$

For example, all of the functions we defined in Lecture 1 can be equivalently expressed as folds:

$$\begin{aligned} nodes &= foldBT\ 0\ (\backslash x\ m\ n \rightarrow 1 + m + n) \\ leaves &= foldBT\ 1\ (\backslash x\ m\ n \rightarrow m + n) \\ height &= foldBT\ 0\ (\backslash x\ m\ n \rightarrow 1 + \max\ m\ n) \\ mirror &= foldBT\ Leaf\ (\backslash x\ tL'\ tR' \rightarrow Node\ x\ tR'\ tL') \end{aligned}$$

## 9 Type classes

By now we’ve seen several examples of polymorphic functions with type class constraints, such as

$$\begin{aligned} sort &:: Ord\ a \Rightarrow [a] \rightarrow [a] \\ lookup &:: Eq\ a \Rightarrow a \rightarrow [(a, b)] \rightarrow Maybe\ b \\ sum, prod &:: Num\ a \Rightarrow [a] \rightarrow a \end{aligned}$$

and so on. Intuitively, these constraints express the minimal requirements on the type  $a$  needed in order to define these functions – for example, *sort* requires a comparison operation on elements, *lookup* requires an equality test, and *sum* and *prod* require some notion of “number” equipped with addition/multiplication operations.

Formally, a type class is defined by specifying the type signatures of operations that need to be implemented to qualify as a member of the class, possibly together with default implementations of some of the operations in terms of the others. For example, the *Eq* class is defined as follows:

```
class Eq a where
  (≡), (≠) :: a → a → Bool
  x ≠ y = not (x ≡ y)
  x ≡ y = not (x ≠ y)
```

This says that to satisfy the *Eq* type class constraint, a type  $a$  needs to at least provide either an equality operation  $(\equiv) :: a \rightarrow a \rightarrow Bool$  or an inequality operation  $(\neq) :: a \rightarrow a \rightarrow Bool$ , and if both are not provided then one is defined as the negation of the other by default. The way that we show that a type class constraint is satisfied is by providing an *instance*. For example, we can define an instance of the *Eq* class for booleans like so:

```
instance Eq Bool where
  False ≡ b = not b
  True ≡ b = b
```

Sometimes we can derive a type class instance for a type constructor if we assume instances for its arguments. For example, two lists can be tested for equality provided that their underlying type of elements supports an equality test:

```
instance Eq a => Eq [a] where
  [] ≡ [] = True
  (x : xs) ≡ (y : ys) = x ≡ y && xs ≡ ys
  _ ≡ _ = False
```

In Haskell it is also possible to express that one type class inherits operations from another type class, in a manner similar to the class hierarchy in object-oriented programming. For example, here is how the *Ord* class is defined in the Prelude (note *Ordering* is a datatype with three values *LT*, *EQ*, and *GT*):

```
class Eq a => Ord a where
  compare :: a -> a -> Ordering
  (<), (≤), (>), (≥) :: a -> a -> Bool
  max, min :: a -> a -> a

  compare x y = if x ≡ y then EQ
               else if x ≤ y then LT
               else GT

  x < y = case compare x y of { LT -> True; _ -> False }
  x ≤ y = case compare x y of { GT -> False; _ -> True }
  x > y = case compare x y of { GT -> True; _ -> False }
  x ≥ y = case compare x y of { LT -> False; _ -> True }

  max x y = if x ≤ y then y else x
  min x y = if x ≤ y then x else y
```

The definition might look a bit complicated, but that's mainly because of all of the default implementations. To provide an instance of *Ord* it suffices to give an implementation of  $(\leq) :: a \rightarrow a \rightarrow \text{Bool}$  or of  $\text{compare} :: a \rightarrow a \rightarrow \text{Ordering}$ , assuming you already have an instance of *Eq*.

It is often implicit in the definition of a type class that the operations obey certain laws. For example, for any instance of the *Eq* class, the binary operation  $(\equiv)$  should be reflexive  $((x \equiv x) = \text{True})$ , symmetric  $((x \equiv y) = (y \equiv x))$ , and transitive (if  $(x \equiv y \ \&\& \ y \equiv z) = \text{True}$  then  $(x \equiv z) = \text{True}$ ): mathematically, we summarize this by saying that  $(\equiv)$  should be an equivalence relation. Similarly, for any instance of *Ord*, the operation  $(\leq)$  should be a linear ordering, i.e., reflexive, transitive, anti-symmetric (if  $(x \leq y \ \&\& \ y \leq x) = \text{True}$  then  $(x \equiv y) = \text{True}$ ), and total  $((x \leq y \ || \ y \leq x) = \text{True})$ . However, although these expectations are sometimes described in the documentation of a type class, it is up to the programmer to respect them when defining an instance of the class, as they are not enforced by the language.<sup>4</sup>

Finally, let's quickly have a look at the *Num* type class:

```
class Num a where
  (+), (-), (*) :: a -> a -> a
```

---

<sup>4</sup>Although they could be, in principle, potentially at the cost of an additional burden of proof on the part of the programmer. In dependently-typed languages such as Coq and Agda it is possible to enforce such properties via the type system.

```

negate      :: a -> a
abs         :: a -> a
signum     :: a -> a
fromInteger :: Integer -> a
x - y      = x + negate y
negate x   = 0 - x

```

The operations  $(+)$ ,  $(-)$ ,  $(*)$  are expected to satisfy the usual arithmetic laws (associativity, distributivity, etc.), while  $signum\ x$  and  $abs\ x$  are meant to return the sign and absolute value, respectively, of the value  $x$ , satisfying the equation  $x = signum\ x * abs\ x$ . The Haskell Prelude defines several instances of the *Num* type class, such as *Num Int*, *Num Integer*, and *Num Double*.

Type classes are a very appealing and useful feature of Haskell, but it is worth mentioning that in a certain sense they may be seen as “just” a convenient mechanism for defining higher-order functions. Indeed, a type class constraint in the type signature of a function may always be replaced by the types of the operations in (a minimal definition of) the corresponding type class, turning these operations into additional arguments to be passed to the function. For example, instead of the first-order function  $lookup :: Eq\ a \Rightarrow a \rightarrow [(a, b)] \rightarrow Maybe\ b$  with a constrained polymorphic type, we can define a higher-order function

```

lookupHO :: (a -> a -> Bool) -> a -> [(a, b)] -> Maybe b
lookupHO eq k [] = Nothing
lookupHO eq k ((k', v) : kvs)
  | eq k k'      = Just v
  | otherwise    = lookupHO eq k kvs

```

taking an equality operation as its first argument. Indeed, its definition is almost exactly the same as the definition of *lookup* we saw in Lecture 1, except that calls to the equality operation ( $\equiv$ ) have been replaced by calls to the first argument *eq*. Similarly, instead of defining a function  $sort :: Ord\ a \Rightarrow [a] \rightarrow [a]$  with a constrained polymorphic type, we can define a higher-order sorting function

```

sortHO :: (a -> a -> Bool) -> [a] -> [a]

```

that expects a comparison operation as its first argument. Then when we want to apply *sortHO* to some list, we simply have to supply this additional comparison operation explicitly.

Still, although this translation of type class constraints into higher-order functions is semantically valid, it does have certain drawbacks. Note that every call to the higher-order functions *sortHO* and *lookupHO* has to pass an extra argument, corresponding to the chosen comparison/equality operation – and while in these examples there is only one additional argument, in general a type class constraint could translate to many extra arguments. One practical benefit of Haskell’s type class mechanism is that these “semantically implicit” arguments are in effect automatically inferred by the type system, by examining the various instances defined within the program and the libraries that it imports. Although that too has its drawbacks – in Haskell it is only possible to define one instance of a type class for a given type, although there may not be a single canonical instance.

For example, the type class *Monoid a* expresses that the type *a* is a monoid, equipped with operations *mappend* :: *a* → *a* → *a* and *mempty* :: *a* that should be associative and unital. But there is no *canonical* way to define an instance *Monoid Integer*; we could for example take *mappend* = (+) and *mempty* = 0, or alternatively *mappend* = (\*) and *mempty* = 1. As a slightly more concrete example, sometimes we may want to compare machine integers for equality modulo 10 or modulo 256 (say), but we can't redefine the *Eq* instance for *Int*. One way around this limitation is to introduce a new type isomorphic to *Int* using the **newtype** mechanism, solely for the purpose of defining new type class instances, e.g.:

```
newtype Int10 = I10 Int
instance Eq Int10 where
  I10 x ≡ I10 y = (mod x 10 ≡ mod y 10)
```

We can then for example write

```
lookup (I10 3) [(I10 2, "a"), (I10 13, "b")]
```

which evaluates to *Just "b"*.